
///////////////////////////////////////////////////////////////
// ‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐//
//                   DK6NF‐CW‐Keyer  V1.00                   //
// ‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐‐//
// Compiler: mikroPascal 3.5 (mikroe.com)                    //
// µC: Atmel AVR ATMEGA 8L8                                  //
//                                                           //
// Oszillator: intern 8 MHz      |   Konfigurationsregister: //
// Brown‐Out‐Detector: On, 4 V   |   LOW  = 0x04 = 0000 0100 //
// Start‐Up‐Time: 6 CK + 64 ms   |   HIGH = 0xD9 = 1101 1001 //
///////////////////////////////////////////////////////////////

program DK6NF‐CW‐Keyer;

var DitFlag, DahFlag : boolean;   // Variablen für "DIT‐" und "DAH‐Flag" definieren
    cwg : dword;                  // Variable für die CW‐Geschwindigkeit definieren
    th : word;                    // Variable für die Mithörtonhöhe definieren

procedure Delay(Zeit_1ms:word);   // Verzögerungsprozedur
var i : word;
begin
  for i := 1 to Zeit_1ms do
    Delay_us (1000);
end;

procedure Dit;            // DIT ausgeben:
begin
  PortD.B7 := 1;          // Sendertastung aktivieren
  Sound_Play(th,cwg);     // Mithörton ausgeben
  PortD.B7 := 0;          // Sendertastung wieder deaktivieren
  Delay (cwg);            // Pause von einem 1 DIT Länge
end;

procedure Dah;            // DIT ausgeben:
begin
  PortD.B7 := 1;          // Sendertastung aktivieren
  Sound_Play(th,cwg*3);   // Mithörton ausgeben ... 1 DAH = 3 DIT
  PortD.B7 := 0;          // Sendertastung wieder deaktivieren
  Delay (cwg);            // Pause von einem 1 DIT Länge
end;

procedure InterruptINT0(); org 0x01;  // Ein Druck auf das DIT‐Paddel löst eine Unterbrechung des Hauptprogramms aus und ...
begin
  DitFlag := true;                    // ... setzt das "DIT‐Flag" auf "gesetzt"
end;

procedure InterruptINT1(); org 0x02;  // Ein Druck auf das DAH‐Paddel löst eine Unterbrechung des Hauptprogramms aus und ...
begin
  DahFlag := true;                    // ... setzt das "DAH‐Flag" auf "gesetzt"
end;

procedure Init;           // Initialisierungsprozedur
begin
  MCUCR := %00001010;     // INT0 und INT1 reagieren auf abfallende Flanke
  GICR  := %11000000;     // INT0 und INT1 freigeben
  DDRD  := %11000000;     // Das Datenrichtungsregister entsprechend der angeschlossenen Hardware setzen
  DitFlag := false;       // Die Voreinstellung für das "DIT‐Flag" ist "nicht gesetzt"
  DahFlag := false;       // Die Voreinstellung für das "DAH‐Flag" ist "nicht gesetzt"
  ADC_Init();             // Analog‐Digital‐Wandler initialisieren
  Sound_Init(PORTD,6);    // Klangfunktion initialisieren
end;

begin                                        // ‐ HAUPTPROGRAMM ‐
  Init;                                      // Initialisierung aufrufen
  Sound_Play(500, 100);                      // Einschaltmeldung, einmal piepsen
  asm sei; end;                              // Interrupts generell freigeben
  while (true) do                            // Endlosschleife ab hier ..........
    begin
      cwg := 30 + ADC_Get_Sample(0) div 10;  // Konstante + Potiwert (0..1023) ergibt den Wert für die CW‐Geschwindigkeit
      th := 400 + ADC_Get_Sample(1);         // Konstante + Potiwert (0..1023) ergibt den Wert für die Mithörtonhöhe



      if DitFlag then                        // Wurde das "DIT‐Flag" durch einen Tastendruck gesetzt? Dann ...
        begin
          Dit;                               // ... ein DIT ausgeben
          if PIND.B2 then                    // Ist das DIT‐Paddel immer noch gedrückt? ‐ Dann ...
            DitFlag := false;                // ... ein weiteres DIT ausgeben
        end;
      if DahFlag then                        // Wurde das "DAH‐Flag" durch einen Tastendruck gesetzt? Dann ...
        begin
          Dah;                               // ... ein DAH ausgeben
          if PIND.B3 then                    // Ist das DAH‐Paddel immer noch gedrückt? ‐ Dann ...
            DahFlag := false;                // ... ein weiteres DAH ausgeben
        end;
    end;
end.


